Lecture 7:

The Wave Equation & Separation of Variables

Mihaela B. Vajiac and Juan Tolosa

PCMI, Summer 2003

```
> restart: with(plots):
Warning, the name changecoords has been redefined
```

Two-dimensional examples

Rectangular oscillating membrane

```
> a:=Pi:b:=Pi:c:=1:
> u := (m,n,x,y,t) ->
  16*((-1)^m-1)*((-1)^n-1)/n^3/m^3/Pi^2*cos((m^2+n^2)^(1/2)*t)*si
  n(m*x)*sin(n*y):
```

The first harmonic is periodic, with period $\sqrt{2}\,\pi$:

```
> animate3d(u(1,1,x,y,t),x=0..a,y=0..b,t=0..Pi*sqrt(2),
  axes=framed,thickness=2, frames=16);
```



$u(1, 3, x, y, t)$ has period $\dfrac{2\,\pi}{\sqrt{10}}$:

```
> animate3d(u(1,3,x,y,t),x=0..a,y=0..b,t=0..2*Pi/sqrt(10),axes=fr
  amed,thickness=2, frames=16);
```



u(5, 3, x, y, t) has period $\frac{2\pi}{\sqrt{34}}$:

```
> animate3d(u(5,3,x,y,t),x=0..a,y=0..b,t=0..2*Pi/sqrt(34),axes=framed,thickness=2, frames=16);
```



Since the various harmonics have different frequencies, a linear combination will not be periodic, but *almost periodic.*

```
> combination := u(1,1,x,y,t) + 10*u(1,3, x,y,t)+
  200*u(5,3,x,y,t):
> animate3d(combination,x=0..a,y=0..b,t=0..1.35*Pi,axes=framed,th
  ickness=2, frames=16);
```




```
>
```

This is a round membrane:

```
> restart: with(plots):
> upartic:= (r, theta, t) ->
  3.583422770*3^(1/2)/((1.913229428*Pi+105.4984657)*Pi)^(1/2)*cos
  (2.567811151*t)*BesselJ(2,5.135622302*r)*sin(2*theta):
> Tperiod := 2*Pi/2.567811151:
> addcoords(z_cylindrical,[z,r,theta],
  [r*cos(theta),r*sin(theta),z]):
> animate3d(upartic(r,theta,t),r=0..1,theta=0..2*Pi,t=0..Tperiod,

  coords=z_cylindrical,axes=BOXED, frames = 12);
```



Damped string

Let us consider a homogeneous problem corresponding to the damped string:

PDE: $u_{tt} = c^2\, u_{xx} - \gamma\, u_t$, where $c^2 = \frac{T}{\rho}$ and γ is a damping factor (positive);

BC: $\mathrm{u}(0, t) = 0$,

$\mathrm{u}(l, t) = 0$;

IC: $\mathrm{u}(x, 0) = \mathrm{f}(x)$,

$u_t(x, 0) = \mathrm{g}(x)$.

We seek non trivial solutions (**eigenfunctions**), using the method of separation of variables, as $u_n(x, t) = \mathrm{X}(x)\,\mathrm{T}(t)$.

As before, the given initial conditions yield

$X_n(x) = \sin\left(\frac{n\,\pi\,x}{l}\right)$, $n = 1, 2, ...$

The time factor is solved next, giving

$T_n(t) = \mathbf{e}^{\left(-\frac{\gamma\,t}{2}\right)}\,(A_n \cos(\alpha_n\, t) + \sin(\alpha_n\, t))$,

where

$\alpha_n = \sqrt{\left(\frac{c\,n\,\pi}{l}\right)^2 - \frac{\gamma^2}{4}}$,

assuming that γ is sufficiently small, so the expression under the radical is positive for all n = 1, 2, ...

The general solution of the whole problem (including the initial conditions) is sought as a linear combination of the eigenfunctions.

Let us consider a particular case.

```
> restart:with(plots):
Warning, the name changecoords has been redefined
```

```
> l:= 1:   c:= 1:  rho:= 1:   T:= 1:
> unprotect(gamma): gamma:= 2:
> assume(n::integer);
> alpha := n -> sqrt((c*n*Pi/l)^2 - gamma^2/4):
```

Initial position:

```
> f := x -> x*(1-x):
```

Initial velocity:

```
> g := x -> 0 :
> A:= n -> (2/l)*int(f(x)*sin(n*Pi*x/l), x=0..l):
> B:= n -> (2/alpha(n)/l)*int((gamma*f(x)/2+g(x))*
            sin(n*Pi*x/l), x=0..l):
```

Eigenfunctions:

```
> upart := (n,x,t) -> exp(-gamma*t/2)*(A(n)*cos(alpha(n)*t)
                         + B(n)*sin(alpha(n)*t))*sin(n*Pi*x/l):
> animate(upart(1,x,t), x=0..1, t=0..4, frames=36, thickness=2);
```




```
> plot3d(upart(1,x,t), x=0..1, t=0..4, orientation=[-15,77]);
```



The plucked string

```
> restart: with(plots):
```

PDE: $u_{tt} = c^2\, u_{xx}$

BC: $u(0, t) = 0,\ u(L, t) = 0.$

IC: $u(x, 0) = f(x), \quad u_t(x, 0) = 0.$

Eigenvalues:

```
> lambda := n -> n*Pi/l:
```

Initial position: $f(x)$is a triangle (plucked string) .
We assume that at a point *p* on (0, l) the string is lifted to height *h*,
remaining fixed at the endpoints.

```
> assume(0<p, p<l, 0<h);
> f := x -> piecewise(x <=p, h*x/p, x > p, h*(l-x)/(l-p)):
```

We plot the function for particular values of the parameters:

```
> particular := {l=1, h=1/2, p=2/3, c=1}:
> plot(subs(particular,f(x)), x=0..1,y=0..1);
```



Fourier coefficientes, A(n).

```
> Af := n -> (2/l)*int(f(x)*sin(lambda(n)*x),x=0..l):
> result :=subs({cos(n*Pi)=(-1)^n, sin(n*Pi)=0}, Af(n)):
> A := unapply(result,n):
```

We find the amplitudes for particular values of the parameters:

```
> AA := subs(particular,unapply(result,n)):
```

This is the solution (eigenfunction) por the particular values we chose:

```
> u:= (n, x, t) -> (AA(n)*cos(c*lambda(n)*t))
                   *sin(lambda(n)*x):
```

Approximate solution, for our particular choice of the parameters:

```
> ua := (x,t) -> subs(particular, sum(u(n,x,t), n=1..17)):
> animate(ua(x,t),x=0..1,t=0..2,color=red,thickness=2);
```



This is a plot of the surface u(x, t).

```
> plot3d(ua(x,t), x=0..1, t=0..2);
```



Localized plucking

Localized plucking, to better see the traveling waves.

We give particular values to all parameters:

```
> l:=1:  h:=1/2:  p:=2/3:   c:=1:
  a:= 2/3-0.2:  b:=2/3+0.1:
> fsmall := x -> piecewise(x>=a and x <=p,h*(x-a)/(p-a),
  x >=p and x<=b, h*(x-b)/(p-b)):
> plot(fsmall(x), x=0..1,y=0..1);
```



```
> Asmall := m -> (2/l)*int(fsmall(x)*sin(lambda(m)*x),x=0..l):
> usmall:= (n, x, t) -> (Asmall(n)*cos(c*lambda(n)*t))
                *sin(lambda(n)*x):
```

Approximate solution for this case:

```
> uaa := (x,t) -> sum(usmall(n,x,t), n=1..23):
> animate(uaa(x,t),x=0..1,t=0..2,color=red,thickness=2,
  frames=22);
```



This is a plot of the surface u(x, t).

```
> plot3d(uaa(x,t), x=0..1, t=0..2, grid=[60,60]);
```



```
>
```

Musical instruments

We have already seen the plucked string.
Let us discuss other problems for the string equation, arising from the way musical instruments are played.

Localized impulse

```
> restart:with(plots):
```

If we hit the string with an impulse *K* concentrated at a point *p*,
(say we hit the string with the blade of a knife) then the solution is given by

The nth armonic is given by

```
> uh := (n, x, t) -> 2*K/(Pi*c*rho)*(1/n)*
         sin(Pi*n*p/l)*sin(Pi*n*x/l)*sin(Pi*n*c*t/l):
> particular:= {l=1, c=1, K=1, p=2/3, rho = 1}:
```

Approximate solution to the "impulse start":

```
> uimp:= (x,t) -> sum(uh(n,x,t),n=1..9):
> animate(subs(particular,uimp(x,t)), x=0..1, t=0..2,
  thickness=2);
```




```
> plot3d(subs(particular,uimp(x,t)), x=0..1, t=0..2,
  axes=BOXED);
```



Small flat hammer

```
> restart: with(plots):
```

The initial position is zero, the initial velocity is constant, equal to v_0, on a small interval $(p - \delta, p + \delta)$. The solution is given by

```
> u(x,t):= Sum(u[n](x,t), n=1..infinity):
```

where the nth armonic is given by

```
> uh := (n, x, t) -> 4*v[0]*l/(Pi^2*c)*(1/n^2)*
         sin(Pi*n*p/l)*sin(Pi*n*delta/l)*
         sin(Pi*n*x/l)*sin(Pi*n*c*t/l):
```

```
> particular:= {l=1, c=1, v[0]=1, p=2/3, rho = 1, delta=.1}:
```

Approximate solution to the "impulse start":

```
> uimp:= (x,t) -> sum(uh(n,x,t),n=1..9):
```

```
> animate(subs(particular,uimp(x,t)), x=0..1,
  t=0..2,thickness=2);
```




```
> plot3d(subs(particular,uimp(x,t)), x=0..1, t=0..2,
  axes=BOXED);
```